// File: src/VestedExchange.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

import {IERC20} from "openzeppelin-contracts/contracts/token/ERC20/IERC20.sol";

interface ILlamaPay {
    function depositAndCreate(
        uint amountToDeposit,
        address to,
        uint216 amountPerSec
    ) external;
}

contract VestedExchange {
    uint256 internal constant _PERIOD = 30 days * 8;
    uint256 internal constant _CRV_AMOUNT = 1_373_277e18;
    uint256 internal constant _CNC_AMOUNT = 588_235e18;
    address internal constant _CRV = 0xD533a949740bb3306d119CC777fa900bA034cd52;
    address internal constant _CNC = 0x9aE380F0272E2162340a5bB646c354271c0F5cFC;
    address internal constant _LLAMA_PAY =
        0x577d4051c35A528DAB369fbbD03ab5277Cf9AeC7;

    uint216 internal immutable _AMOUNT_PER_SEC;
    address internal immutable _MICHAEL;

    address public preparer;
    bool public prepared;
    bool public started;

    constructor(address michael_) {
        _MICHAEL = michael_;
        // Multiplying by 1e2 as Llama pay expects 20 decimals
        _AMOUNT_PER_SEC = uint216(_CNC_AMOUNT / _PERIOD) * 1e2;
    }

    // Called by Conic to prepare the vesting (first step)
    function prepare() external {
        require(!prepared, "VestedExchange: vesting already prepared");
        IERC20(_CNC).transferFrom(msg.sender, address(this), _CNC_AMOUNT);
        prepared = true;
        preparer = msg.sender;
    }

    // Cancel the prepared vesting (only used if the vesting is not started)
    function cancel() external {
        require(prepared, "VestedExchange: vesting not prepared");
        require(msg.sender == preparer, "VestedExchange: not preparer");
        require(!started, "VestedExchange: vesting already started");
        IERC20(_CNC).transfer(msg.sender, _CNC_AMOUNT);
        delete prepared;
        delete preparer;
    }

    // Called by Michael to start the vesting (second and final step)
    function start() external {
        require(msg.sender == _MICHAEL, "VestedExchange: not michael");
        require(prepared, "VestedExchange: vesting not prepared");
        require(!started, "VestedExchange: vesting already started");
        IERC20(_CRV).transferFrom(msg.sender, preparer, _CRV_AMOUNT);
        IERC20(_CNC).approve(_LLAMA_PAY, _CNC_AMOUNT);
        ILlamaPay(_LLAMA_PAY).depositAndCreate(
            _CNC_AMOUNT,
            _MICHAEL,
            _AMOUNT_PER_SEC
        );
        started = true;
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}